perltutorial
stonecolddevin
<p>
You've read the [http://search.cpan.org/%7Eflora/Catalyst-Runtime-5.80004/lib/Catalyst/DispatchType/Chained.pm|tutorial] but you still can't seem to get a grasp on what exactly to use Chained dispatching for.
</p><p>
Chained is just another method of method/url dispatching that Catalyst allows you to use. Chained, however, allows you to chain together methods that need to share information across requests....
</p><p>
Say you want to construct a url that looks something like this: http://www.yourhost.com/item/1/view. Normally, you'd have to go through a bunch of parsing involving <strike>$ENV{'PATH_PART'}</strike> $ENV{'PATH_INFO}, which is ugly, and hackish. You can do it like this with Chained:
</p><p>
<c>
# step one:
# set up a base method to make sure your item exists, and
# store your item information in $c->stash
sub base : Chained('/') PathPart('item') CaptureArgs(1) {
my ($self, $c, $itemid) = @_;
my $item = $c->model('Item')->find($itemid);
if ( $item == undef ) {
$c->stash( error_msg => "This item does not exist" );
$c->detach("nonexistent");
} else {
$c->stash( item => $item );
}
}
</c>
</p>
<p>
Let's break this method down, and then we'll get to the rest of this chain.
</p><p><c>
sub base : Chained('/') PathPart('item') CaptureArgs(1)
</c><p>
the Chained('/') attribute sets up the base of the chain.
</p><p>
PathPart determines the name of the path to capture.
</p><p>
CaptureArgs tells Chained to capture one argument directly after the path part.
</p><p>
Now, let's finish the chain so we can have a working URL.
</p><p><c>
## the next part is to link the end point to the chain
## thisis done by passing the name of the base chain method to Chained for this method
sub view : Chained('base') PathPart('view') Args(0) {
my ($self, $c) = @_;
my $item = $c->stash->{item};
$c->stash(
template => 'item/view.tt2',
item => $item,
);
}</c>
</p>
This method is fairly self explanatory, but just to make sure, we'll go ahead and go over it.
</p><p>
As in the comments, Chained is passed the name of the base point of the chain, which in this case is 'base'. We determine the path part with PathPart once again, which can be noted that it doesn't have to match the current method's name. The one different attribute is the Args(0) attribute. Args tells Chained that the end of the chain has been reached, thus giving you a usable URL at this point. You can specify arguments to be passed in Args(), and they can be accessed via the @_ variable, like $self and $c.
</p><p>
To recap, this gives us a url like http://www.yourdomain.com/item/itemid/view. You can expand this for CRUD actions so you can have URLs like the above, as well as something like /item/create. Here is some example code:
</p>
<c>
sub blog : Chained('/') PathPart('blog') CaptureArgs(0) {
}
sub load_blog : Chained('blog') PathPart('') CaptureArgs(1) {
my ($self, $c, $id) = @_;
my $entry = $c->model('DB::Blog')->find($id);
if ( $entry ) {
$c->stash(entry => $entry );
} else {
$c->stash( error_msg => "No such blog" );
$c->detach;
}
}
sub view : Chained('load_blog') PathPart('view') Args(0) {
my ($self, $c) = @_;
}
sub create : Chained('blog') PathPart('new') Args(0) FormConfig('blogs/create.yml') {
my ($self, $c) = @_;
my $form = $c->stash->{form};
if ( $form->submitted_and_valid ) {
my $entry = $c->model('DB::Blogs')->new_result({ created => DateTime->now });
$form->model->update($entry);
$c->stash(
status_msg => "Post saved",
entry => $entry
);
} else {
$c->stash( error_msg => "Your form had errors" );
$c->detach;
}
}
</c>
<p>
This actually creates a couple different Chained instances, giving us the ability to have urls like /blog/id/view, blog/new, etc.
</p>
<p>
Next time, using $c->uri_for_action to create URIs for Chained methods.
</p><!-- Node text goes above. Div tags should contain sig only -->
<div class="pmsig"><div class="pmsig-203787">
<p>mtfnpy</p>
</div></div>